

Article Title: "SPICE Models For Power Electronics"

Author: L.G. Meares and
Charles E. Hymowitz

Abstract: Due to the increasing complexity of power systems and the costs involved in breadboarding and testing preliminary designs, engineers have been turning to computer based simulations for assistance in the design phase.

This paper explains, in depth, how to create SPICE models for the most important and pervasive element in power electronics, that of the transformer and its accompanying saturable core.

Outline:


- I Transformer Models
- II Saturable Reactor Model
- III How The Core Model Works
- IV Calculating Core Parameters
- V Using and Testing The Core
- VI A Complete Transformer Model
- VII References

Total: 11 Pages, 12 Figures


Models For Power Electronics

Transformer Models

The usual method of simulating a transformer using IsSPICE is by specifying the open circuit inductance seen at each winding and then adding the coupling coefficients to a pair of coupled inductors. This technique tends to lose the physical meaning associated with leakage and magnetizing inductance and does not allow the insertion of a nonlinear core. It does, however, provide a transformer that is simple to create and simulates efficiently. The coupled inductor type of transformer, its related equations and relationship to an ideal transformer with added leakage and magnetizing inductance is shown in Figure 1.



In order to make a transformer model that more closely represents the physical processes, it is necessary to construct an ideal transformer and model the magnetizing and leakage inductances separately. The ideal transformer is one that preserves the voltage and current relationships, shown in Figure 2, and has a unity coupling coefficient and infinite magnetizing inductance. The ideal transformer, unlike a real transformer, will operate at DC. This property is useful for modeling the operation of DC-DC converters.




The coupling coefficient of a transformer wound on a magnetic core is nearly unity when the core is not saturated and depends on the winding topology when the core is saturated. The work of Hsu, Middlebrook and Cuk [3] develops the relationship of leakage inductance, showing that relatively simple measurements of input inductance with shorted outputs yield the necessary model information.

The IsSPICE equivalent circuit for an ideal transformer is shown in Figure 3 and implements the following equations:


$$V1 * \text{RATIO} = V2$$

$$I1 = I2 * \text{RATIO}$$

RP and RS are used to prevent singularities in applications where terminals 1 and 2 are open circuited or terminals 3 and 4 are connected to a voltage source. RATIO is the turns ratio from winding 3,4 (secondary) to winding 1,2 (primary). Polarity “dots” are on terminals 1 and 3 as shown in Figure 2.



Multi-winding topologies can also be simulated by using combinations of this 2 port representation.



Now that the ideal transformer has been constructed, magnetizing inductance can be added using a separate saturable core model described next.

Saturable Reactor Model

A saturable reactor is a magnetic circuit element consisting of a single coil wound around a magnetic core. The presence of a magnetic core drastically alters the behavior of the coil by increasing the magnetic flux and confining most of the flux to the core. The magnetic flux density, B , is a function of the applied MMF, which is proportional to ampere turns. The core consists of a number of tiny magnetic domains made up of magnetic dipoles. These domains set up a magnetic flux that adds to or subtracts from the flux set up by the magnetizing current. After overcoming initial friction, the domains rotate like small DC motors, to become aligned with the applied field. As the MMF is increased, the domains rotate one by one until they are all in alignment and the core saturates. Eddy currents are induced as the flux changes, causing added loss.

The saturable reactor cannot be modeled using a single SPICE primitive element. Therefore, a saturable core “macro model”, utilizing the `IsSPICE` subcircuit feature, must be created. The saturable core model is capable of simulating nonlinear transformer behavior including saturation, hysteresis, and eddy current losses. To make the model even more useful it has been parameterized. This is a technique which allows the characteristics of the core to be determined just by the specification of a few key parameters. At the time of the simulation, the specified parameters are passed into the subcircuit. The equations in the subcircuit (inside the curly braces) are then evaluated and replaced with a value making the equation based subcircuit compatible with any SPICE program.


The parameters that must be passed to the subcircuit include:

- Flux Capacity in Volt-Sec (VSEC)**
- Initial Flux Capacity in Volt-Sec (IVSEC)**
- Magnetizing Inductance in Henries (LMAG)**
- Saturation Inductance in Henries (LSAT)**
- Eddy current critical frequency in HZ (FEDDY).**

The saturable core may be added to the model of the ideal transformer to create a more complete transformer model. To use the saturable core model just place the core across the transformer's input terminals and evaluate the equations in curly braces. A special subcircuit test point has been provided to allow the monitoring of the core flux. Since there are two connections in the subcircuit, no connection need be made at the top subcircuit level other than the dummy node number.

A call to the saturable core subcircuit using `IsSPICE` would look like the following:

```
X1 2 0 3 CORE {VSEC=50U IVSEC=-25U LMAG=10MHY LSAT=20UHY FEDDY=20KHZ}
```




```
.SUBCKT CORE 1 2 3
F1 1 2 VM1 1
G2 2 3 1 2 1
E1 4 2 3 2 1
VM1 4 5
RX 3 2 1E12
CB 3 2 {VSEC/500}
+ IC={IVSEC/VSEC*500}
RB 5 2 {LMAG*500/VSEC}
RS 5 6 {LSAT*500/VSEC}
VP 7 2 250
D1 6 7 DCLAMP
VN 2 8 250
D2 8 6 DCLAMP
.MODEL DCLAMP D(CJO={3*VSEC/
+ (6.28*FEDDY*500*LMAG)} VJ=25)
.ENDS
```


Figure 5. To make the netlist for the saturable core subcircuit SPICE compatible, just replace all the equations in the curly braces with numerical values. The key parameters are shown in bold.

How The Core Model Works

Modeling the physical process performed by a saturable core is most easily accomplished by developing an analog of the magnetic flux. This is done by integrating the voltage across the core and then shaping the flux analog with nonlinear elements to cause a current to flow proportional to the desired function. This gives good results when there is no hysteresis as illustrated in Figure 6.



The input voltage, $V(2)$, is integrated using the voltage controlled current source, G , and the capacitor CB . An initial condition across the capacitor allows the core to have an initial flux. The output current from F is shaped as a function of flux using the voltage sources VN and VP and diodes $D1$ and $D2$. The inductance in the high permeability region is proportional to RB , while the inductance in the saturated region is proportional to RS . Voltage VP and VN represent the saturation flux. Core losses



can be simulated by adding resistance across the input terminals; however, another equivalent method is to add capacitance across resistor RB in the simulation. Current in this capacitive element is differentiated in the model to produce the effect of resistance at the terminals. The capacitance can be made a nonlinear function of voltage which results in a loss term that is a function of flux. A simple but effective way of adding the nonlinear capacitance is to give the diode parameter, CJO, a value, as is done here. The other option is to use a nonlinear capacitor across nodes 2 and 6, however, the capacitor's polynomial coefficients are a function of saturation flux, causing their recomputation if VP and VN are changed.

Losses will increase linearly with frequency simulating high frequency core behavior. A noticeable increase in MMF occurs when the core comes out of saturation, an effect that is more pronounced for square wave excitation than for sinusoidal excitation as shown in Figure 8. These model properties agree closely with observed behavior [2]. The model is set up for orthonol and steel core materials which have a sharp transition from the saturated to the unsaturated region. For permalloy cores the transition out of saturation is less pronounced. To account for the different response the capacitance value in the diode model (CJO in DCLAMP), which affects core losses, should be scaled down. Also, scaling the voltage sources VN and VP down will soften the transition.

The DC B-H loop hysteresis, usually unnecessary for most applications, is not modeled because of the extra model complexity, causing a prediction of lower loss at low frequencies. The hysteresis, however, does appear as a frequency dependent function, as seen on the previous page, providing reasonable results for most applications, including magnetic amplifiers. The model shown in Figure 7 simulates the core characteristics and takes into account the high frequency losses associated with eddy currents and transient widening of the B-H loop caused by magnetic domain angular momentum. Losses will increase linearly with frequency, simulating high frequency core behavior.




Figure 8, The saturable core model is capable of being used with both sine (below) and square (above) wave excitation as shown in these IsSPICE simulations.

Calculating Core Parameters

The saturable core model is setup to be described in electrical terms, thus allowing the engineer to design the circuitry first without knowledge of the core's physical makeup. After the design is completed, the final electrical parameters can then be used to calculate the necessary core magnetic/size values. The core model could be altered to take as its input magnetic and size parameters. The core could then be described in terms of N , A_c , MI , μ , and B_m and would be more useful for studying previously designed circuits. But the electrical based model is better suited to the natural design process. The saturable core model's behavior is defined by the set of

electrical parameters, shown in Figure 6 and Figure 9. The core's magnetic/size values can be easily calculated from the following equations which utilize cgs units.

| Parameters Passed To Model | |
|-----------------------------------|--|
| VSEC | Core Capacity in Volt-Sec |
| IVSEC | Initial Condition in Volt-Sec |
| LMAG | Magnetizing Inductance in Henries |
| LSAT | Saturation Inductance in Henries |
| FEDDY | Frequency when LMAG Reactance = Loss Resistance in Hz |
| Equation Variables | |
| Bm | Maximum Flux Density in Gauss |
| H | Magnetic Field Strength in Oersteds |
| Ac | Area of the Core in cm ² |
| N | Number of Turns |
| MI | Magnetic Path Length in cm |
| μ | Permeability |

Faraday's law, which defines the relationship between flux and voltage is:

$$E = N \, d\phi/dt * 10^{-8} \quad \text{Eq. 1}$$

where E is the desired voltage, N is the number of turns and ϕ is the flux of the core in maxwells. The total flux may also be written as:

$$\phi_T = 2 * B_m * A_c \quad \text{Eq. 2}$$

Then, from 1 & 2,

$$E = 4.44 * B_m * A_c * F * N * 10^{-8} \quad \text{Eq. 3}$$

and

$$E = 4.0 * B_m * A_c * F * N * 10^{-8} \quad \text{Eq. 4}$$

where Bm is the flux density of the material in Gauss, Ac is the effective core cross sectional area in cm², and F is the design frequency. Equation 3 is for sinusoidal conditions while equation 4 is for a square wave input. The parameter VSEC can then be determined by integrating the input voltage resulting in:

$$\int e dt = N\phi_T = N * 2 * B_m * A_c * 10^{-8} = \mathbf{VSEC} \quad \text{Eq. 5}$$

also from $E = L di/dt$ we have,

$$\int e dt = Li \quad \text{Eq. 6}$$

The initial flux in the core is described by the parameter **IVSEC**. To use the IVSEC option you must put the UIC keyword in the `ISPIECE ".TRAN"` statement. The relationship between the magnetizing force and current is defined by Ampere's law as

$$H = .4 * \pi * N * i / MI \quad \text{Eq. 7}$$

where H is the magnetizing force in oersteds, i is the current through N turns, and MI is the magnetic path length in cm.


From equations 5, 6, and 7 we have

$$L = N^2 * B_m * A_c * (.4 * \pi * 10^{-8}) / H * MI \quad \text{Eq. 8}$$


with $\mu = B/H$ we have

$$\mathbf{L(mag, sat)} = \mu(\text{mag, sat}) * N^2 * .4 * \pi * 10^{-8} * A_c / MI \quad \text{Eq. 9}$$

The values for LMAG and LSAT can be determined by using the proper value of μ in Eq. 9. The values of permeability can be found by looking at the B - H curve and choosing two values for the magnetic flux, one value in the linear region where the permeability will be maximum and one value in the saturated region. Then, from a curve of permeability versus magnetic flux, the proper values of μ may be chosen. The value of μ in the saturated region will have to be an average value over the range of interest. The value of FEDDY, the eddy current critical frequency, can be determined from a graph of permeability versus frequency, shown in Figure 4. By choosing the approximate 3 db point for μ , the corresponding frequency can be determined.



Using And Testing The Saturable Core



The test circuit shown in Figure 10 can be used to evaluate the saturable core model. Pass the core parameters in the curly braces into the saturable core subcircuit and adjust the voltage levels in the "V2 4 0 PULSE" or "V2 4 0 SIN" statements to insure that the core will saturate. You can use Eq. 3 and 4 to get an idea of the voltage levels necessary to saturate the core. The .TRAN statement may also need adjustment depending on the frequency specified by the V2 source. The core parameters must remain reasonable or the simulation may fail. After the simulation is complete, plotting V(5) versus I(VM1) (Flux vs. Current through the core) will result in a B-H plot.

```

Test Circuit
.OPTIONS LIMPTS=1000
*SPICE_NET
.TRAN .1US 50US
*INCLUDE DEVICE.LIB
.OPTIONS LIMPTS=1000
*ALIAS V(3)=VOUT
*ALIAS V(5)=FLUX
*ALIAS V(4)=VIN
.PRINT TRAN V(3) V(5) I(VM1) V(4)
R1 4 2 100
R2 3 0 50
X1 1 0 5 CORE {VSEC=25U IVSEC=-25U
+ LMAG=10MHY LSAT=20UHY FEDDY=25KHZ}
X2 2 0 3 0 XFMR {RATIO=.3}
VM1 2 1
V2 4 0 PULSE -5 5 0US 0NS 0NS
+25US
*Use the Pulse statement for square wave excitation
*V2 4 0 SIN 0 5 40K
*Use the Sin statement for sine wave excitation
*Adjust voltage levels to insure core saturation
.END

```

Figure 11, Equivalent IsPICE netlist for the saturable core test circuit.

A Complete Transformer Model

The magnetizing inductance is added by using the saturable reactor model across any one of the windings of the ideal transformer. Coupling coefficients are inserted in the model by adding the series leakage inductance for each winding as shown in Figure 12.




Figure 12, A Complete Transformer Model. The saturable core may be combined with the ideal transformer, XFMR, and some leakage inductance and series resistance to create a complete model of a transformer.

The leakage inductances are measured by finding the short circuit input inductance at each winding and then solving for the individual inductance. These leakage inductances are independent of the core characteristic shown by ref [3]. The final model, incorporating the CORE and XFMR subcircuits along with the leakage inductance and winding resistance is shown in Figure 12.

SPICE models cannot represent all possible behavior because of the limits of computer memory and run time. This model, as most simulations, does not represent all cases.

Modeling the core (in Figure 12) as a single element referred to one of the windings works in most cases; however, some applications may experience saturation in a small region of the core, causing some windings to be decoupled faster than others, invalidating the model. Another limitation of this model is for topologies with magnetic shunts or multiple cores. Applications like this can frequently be solved by replacing the single magnetic structure with an equivalent structure using several transformers, each using the model presented here.

References

- [1] SPICE2, A COMPUTER PROGRAM TO SIMULATE SEMICONDUCTOR CIRCUITS
Laurence W. Nagel, Memorandum No. ERL-M520, 9 May 1975, Electronics Research Laboratory, College of Engineering, University of California, Berkeley, CA 94720
- [2] DESIGN MANUAL FEATURING TAPE WOUND CORES,
Magnetics, Inc., Components Div., Box 391, Butler PA 16001
- [3] TRANSFORMER MODELING AND DESIGN FOR LEAKAGE CONTROL
Shi Ping Hsu, R.D. Middlebrook and Slobodan Cuk, Power conversion International, pg. 68, Feb, 1982
- [4] ADVANCES IN SWITCHED-MODE POWER CONVERSION
Slobodan Cuk and R.D. Middlebrook, vol. 2, copyright 1981
- [5] NEW SIMULATION TECHNIQUES USING SPICE
L.G. Meares, Applied Power Electronics Conference, (c) IEEE, April-May, 1986.